

Description

`range` generates a numerical range, which is useful for evaluating and graphing functions.

Quick start

Generate `newv1` that ranges from 0 to π

```
range newv1 0 _pi
```

Same as above, but only for the first 50 observations in the dataset

```
range newv1 0 _pi 50
```

Generate `newv2` that ranges from the minimum to the maximum of `v2` after `summarize`

```
range newv2 r(min) r(max)
```

Menu

Data > Create or change data > Other variable-creation commands > Generate numerical range

Syntax

```
range varname #first #last [ #obs ]
```

Remarks and examples

`range` constructs the variable *varname*, taking on values $\#_{\text{first}}$ to $\#_{\text{last}}$, inclusive, over $\#_{\text{obs}}$. If $\#_{\text{obs}}$ is not specified, the number of observations in the current dataset is used.

`range` can be used to produce increasing sequences, such as

```
. range x 0 12.56 100
```


or it can be used to produce decreasing sequences:

```
. range z 100 1
```

► Example 1

To graph $y = e^{-x/6}\sin(x)$ over the interval $[0, 12.56]$, we can type


```
. range x 0 12.56 100
Number of observations (_N) was 0, now 100.
. generate y = exp(-x/6)*sin(x)
. scatter y x, yline(0) ylabel(y = exp(-x/6) sin(x))
```



► Example 2

Stata is not limited solely to graphing functions—it can draw parameterized curves as well. For instance, consider the curve given by the polar coordinate relation $r = 2 \sin(2\theta)$. The conversion of polar coordinates to parameterized form is $(y, x) = (r \sin \theta, r \cos \theta)$, so we can type

```
. clear
. range theta 0 2*_pi 400
Number of observations (_N) was 0, now 400.
. generate r = 2*sin(2*theta)
. generate y = r*sin(theta)
. generate x = r*cos(theta)
. line y x, c(1) m(i) yline(0) xline(0) aspectratio(1)
```



◀

Also see

[D] [egen](#) — Extensions to generate

[D] [obs](#) — Increase the number of observations in a dataset

Stata, Stata Press, and Mata are registered trademarks of StataCorp LLC. Stata and Stata Press are registered trademarks with the World Intellectual Property Organization of the United Nations. StataNow and NetCourseNow are trademarks of StataCorp LLC. Other brand and product names are registered trademarks or trademarks of their respective companies. Copyright © 1985–2025 StataCorp LLC, College Station, TX, USA. All rights reserved.

For suggested citations, see the FAQ on [citing Stata documentation](#).

